2次元ディスタンスフィールドまとめ

# はじめに

[iq先生の2Dディスタンスフィールド(SDF)解説](http://www.iquilezles.org/www/articles/distfunctions2d/distfunctions2d.htm) を見ましょう。3Dの方は有名なのですが、こっち見逃してました。。。もっと早く知りたかった

# ディスタンスフィールド(距離関数)とは
相手との距離を返す関数です。
glslのフラグメントシェーダーでシェイプを描くのに利用されます。

# ディスタンスフィールド(距離関数)の種類

0以上か、マイマスがあるのかの、符号あり、なしの2種類あるのですが、全く用途が異なります。図形を描く場合はもっぱらSDFが使われるので、だいたいSDFを指しますが、USDFじゃないとダメな場合があるので注意が必要です。

* 符号なしディスタンスフィールド(unsigned distance function 略してUSDF?)

 原点からの距離を測るものなので絶対にマイナスになりません。ボロノイの距離関数として使えます。有名なのはマンハッタン距離を使った回路みたいな模様を作るボロノイですね。USDFはなんでも使えるので、多種多様なボロノイを作ることが出来ます。また、ディスプレイスメント、極座標変換にも利用できます。length関数になっているところは、すべて置き換え可能なので、実はかなり使えます。この関数単体では図形になりませんが、相手との距離を設定する(引く)ことで図形になります。これが後述するSDFです。

* 符号ありディスタンスフィールド(signed distance function 略してSDF)

 基本的には、USDFからある距離を引いたものになりますが、図形の辺を原点としている為、必ずしもそうではありません。(原点が1つの点にならない場合がある)なので、SDFをUSDFとして戻して使うことができない場合が多々あります。

左上 USDF 右上 SDF 相互変換できる場合(原点が1つの点)
![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/135381/a61243c2-8ea8-296e-b1f5-af69b117f5c1.png)
左下 USDF 右下 SDF 相互変換できない場合(原点が縦軸の直線)

符号ありディスタンスフィールドは2種類あり、境界線から、内側外側のどちらがプラスかマイナスかの二種類ある。結論から言うと、境界線より外側をプラスとするiq先生式が使いやすい。境界線より内側をプラスとする方は、図形の形をはっきり認識できるので、理解するにはこちらの方がいいが、グローとかを掛けるときに不具合が生じる。

|ベクトル|距離(スカラー)|
|---|---|
|ベクトルフィールド(p)|符号なしディスタンスフィールド(length(p))|
|符号ありベクトルフィールド(normalize(p)-p)|符号ありディスタンスフィールド(length(p)-1)|
|符号ありベクトルフィールド(p-normalize(p))|符号ありディスタンスフィールド(1.-length(p))|

符号ありベクトルフィールドは、符号ありディスタンスフィールドのベクトル版、ベクトルフィールドのベクトルの方向がそのまま法線となる為、微分して法線ベクトルを計算する必要はない。またベクトルデータが保存されているため、ベクトルフィールドを多段で掛けることが出来る。符号ありディスタンスフィールドへの変換はlengthを掛けるだけ。※虚数の補正は必要

# まずは基本のlength3兄弟

## 円(length)

![length.png](https://qiita-image-store.s3.amazonaws.com/0/135381/628489ef-4624-e0ad-a3a9-cdf22ec8c1f5.png)
https://goo.gl/KxQ9Rq

```glsl
float l=length(p);
```

これは、いわゆるユークリッド(EUCLID)距離です。L2距離とも呼ばれています。

```glsl
l=sqrt(p.x*p.x+p.y*p.y);
```

シンプルに、相手との距離を測る関数lengthを使うだけです。もちろん、それは円状になりますね。これがすべての基本になります。

後述のlengthN関数の中ではlength2に相当します。

length(p)と同じように円を描く距離関数にdot(p,p)がありますが、

![dot.png](https://qiita-image-store.s3.amazonaws.com/0/135381/a14a99af-00e0-5c62-3626-ef952c6cadda.png)
https://goo.gl/ZBLHrm

```glsl
float l=dot(p,p);
```

こちらは、半径1の円を描くときには同じですが、等間隔ではないので注意が必要です。
球状の塗りをする場合にはこちらを使います。length(p)の場合は円錐状の塗りになります。
length(p)よりも軽いという特徴があります。ベタ塗りだったらdotの方が良いです。

ちなみに、これはユークリッド距離のsqrtを取ったものと同じです

```glsl
l=p.x*p.x+p.y*p.y;
```

## 正方形

45度回転バージョンと通常バージョンの2パターンあります。

lengthを使っていませんが、ようは、X軸、Y軸に沿って距離を図るということですね。

* まずは通常バージョン

![square.png](https://qiita-image-store.s3.amazonaws.com/0/135381/671408c7-f1d8-1a83-0991-24d9d6dffa6f.png)


https://goo.gl/VMcZwS

```glsl
precision mediump float;
uniform vec2  m;       // mouse
uniform float t;       // time
uniform vec2  r;       // resolution
uniform sampler2D smp; // prev scene

void main(void){
	vec2 p = (gl_FragCoord.xy * 2.0 - r) / min(r.x, r.y);
	float l=max(abs(p.x),abs(p.y));
	gl_FragColor = vec4(vec3(l), 1.0);
}
```

これは後述のlengthN関数の中ではlengthInfに相当します。
Nを無限にした場合ですね。 L∞距離とも呼ばれています。
チェビシェフ距離とも呼ばれています。
関数としてはmaxabs関数です

* 正方形 45度バージョン L1距離

これは、[マンハッタン距離](https://mathtrain.jp/manhattan)です。

![square2.png](https://qiita-image-store.s3.amazonaws.com/0/135381/6fb37037-f8f9-7c14-fab7-1071ee6ff518.png)
https://goo.gl/sKx4FS

```glsl
precision mediump float;
uniform vec2  m;       // mouse
uniform float t;       // time
uniform vec2  r;       // resolution
uniform sampler2D smp; // prev scene

void main(void){
	vec2 p = (gl_FragCoord.xy * 2.0 - r) / min(r.x, r.y);
	float l=abs(p.x)+abs(p.y);
	gl_FragColor = vec4(vec3(l), 1.0);
}
```
これは後述のlengthN関数の中ではlength1に相当します。
関数としてはsumabs関数ですね

## 円と正方形の間 lengthN

乗数Nが増えると円の形状から段々正方形に近くなります。

尚、角丸正方形とは異なり、全体的に丸くなります。

* N<1 星型
* N=1 45度の正方形
* N=2 円
* N>2 円から正方形に近づく

イラストレーター的には、パンクと膨張です。

* [Nによる図形の変化](https://goo.gl/D73XCz)

### N=1で45度の正方形

![square2.png](https://qiita-image-store.s3.amazonaws.com/0/135381/6fb37037-f8f9-7c14-fab7-1071ee6ff518.png)
https://goo.gl/2rgudQ

```glsl
precision mediump float;
uniform vec2  m;       // mouse
uniform float t;       // time
uniform vec2  r;       // resolution
uniform sampler2D smp; // prev scene

float lengthN(vec2 v, float n){
  vec2 tmp = pow(abs(v), vec2(n));
  return pow(tmp.x+tmp.y, 1.0/n);
}

void main(void){
	vec2 p = (gl_FragCoord.xy * 2.0 - r) / min(r.x, r.y);
	float l=lengthN(p,1.);
	gl_FragColor = vec4(vec3(l), 1.0);
}
```

### N=2で円

![length2.png](https://qiita-image-store.s3.amazonaws.com/0/135381/703b3008-7256-fcb5-0a8d-2209557cf427.png)
https://goo.gl/xKEh3B

つまりはlength関数と一緒です。length2 = length

### Nが2より大きい(6)とNが大きいほど正方形に近づく

![length6.png](https://qiita-image-store.s3.amazonaws.com/0/135381/83351bfc-93b3-5938-0aed-bdc9858200c1.png)
https://goo.gl/tzkopj

```glsl
	float l=lengthN(p,6.);
```

### Nが1より小さい(.5)とアステロイド

![length05.png](https://qiita-image-store.s3.amazonaws.com/0/135381/98f9cf08-a573-a4d9-8169-782d78c7389c.png)
https://goo.gl/NpGq49

```glsl
	float l=lengthN(p,.5);
```

### Nが無限(Infinite)だとと正方形
![square.png](https://qiita-image-store.s3.amazonaws.com/0/135381/671408c7-f1d8-1a83-0991-24d9d6dffa6f.png)

といっても

```glsl
	float l=lengthN(p,1./0.);
```

とかして、無限を入れるとシェーダーがエラーも吐かず動作しない可能性が高いので辞めましょうね。※ 環境によっては動作するかもしれません。

```glsl
	float l=lengthN(p,1e20);
```

代わりに無限に近い巨大な数1e20 (= 10^20 = 一垓)と入れるとほぼ正方形になります。
まぁ、計算量が無駄ですので、素直に、正方形の描き方の方を使いましょう。

# 十字

maxがN∞だったらminはN-∞

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/135381/eb680f33-17c6-2e00-2807-e739739dda3c.png)

```glsl
	float l=min(abs(p.x),abs(p.y));
```

尚powは重いのでNの数が決まっているのであれば専用のlength6やlength8を作って使っているパターンもあります。

```glsl
float length6( vec2 p ){
	p = p*p*p; p = p*p;
	return pow( p.x + p.y, 1.0/6.0 );
}

float length8( vec2 p ){
	p = p*p; p = p*p; p = p*p;
	return pow( p.x + p.y, 1.0/8.0 );
}
```

これは[ノルム](https://mathtrain.jp/lpnorm)(lpnorm)です。
[ミンコフスキー距離](https://en.wikipedia.org/wiki/Minkowski_distance)とも言われています。

|ノルム(n)|図形|数学的距離名|関数名|
|---|---|---|---|
|-∞ |十字        |- |minabs|
|.5 |アステロイド |-  ||
|1  |正方形(45度)|マンハッタン距離(L1距離)|sumabs|
|2  |円          |ユークリッド距離(L2距離)|length|
|∞  |正方形      |チェビシェフ距離|maxabs|
|n  |-          |ミンコフスキー距離|lpnorm(lengthN)|

* [3次元レイマーチングでの距離関数の実装はこちらが参考になります](https://qiita.com/muripo_life/items/e1c2167eb191cf85bcf5)


## 多角形

### 多角形

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/135381/ae2e1eb1-309b-5540-a200-eb1cf84f1924.png)

https://goo.gl/AdfFVm

```glsl
precision mediump float;
uniform vec2  m;       // mouse
uniform float t;       // time
uniform vec2  r;       // resolution
uniform sampler2D smp; // prev scene

const float PI  = acos(-1.);
const float TAU = PI * 2.;

float lPoly(vec2 p,float n){
  float a = atan(p.x,p.y)+PI;
  float r = TAU/n;
  return cos(floor(.5+a/r)*r-a)*length(p)/cos(r*.5);
}

void main(void){
	vec2 p = (gl_FragCoord.xy * 2.0 - r) / min(r.x, r.y);
	float l=lPoly(p,3.);
	gl_FragColor = vec4(vec3(l), 1.0);
}
```


### 多角形星

![starpolygon.png](https://qiita-image-store.s3.amazonaws.com/0/135381/cfeb7294-c3f4-9f6e-16e7-cf99c56614b8.png)

https://goo.gl/xHVUqP

```glsl
precision mediump float;
uniform vec2  m;       // mouse
uniform float t;       // time
uniform vec2  r;       // resolution
uniform sampler2D smp; // prev scene

const float PI = 3.14159265359;
const float PI2 = PI *2.;

float lPolygon(vec2 p,int n){
 float a = atan(p.x,p.y)+PI;
 float r = PI2/float(n);
 return cos(floor(.5+a/r)*r-a)*length(p);
}

mat2 mRotate(float a){
 float c=cos(a);
 float s=sin(a);
 return mat2(c,-s,s,c);
}

float lStarPolygon(vec2 p,int n,float o){
 return (lPolygon(p,n)-lPolygon(p*mRotate(PI2/float(n)/2.),n)*o)/(1.-o);
}

void main(void){
	vec2 p = (gl_FragCoord.xy * 2.0 - r) / min(r.x, r.y);
	float l = lStarPolygon(p,5,.5);
	gl_FragColor = vec4(vec3(l), 1.0);
}
```

### 多角形カーブ

多角形をパンクさせたり、収縮させたような図形を描く-でパンク+で収縮

![polycurve.png](https://qiita-image-store.s3.amazonaws.com/0/135381/d1e4c7ff-ce30-9edc-344c-851219a9d3c6.png)

https://goo.gl/NegXtW

```glsl
precision mediump float;
uniform vec2  m;       // mouse
uniform float t;       // time
uniform vec2  r;       // resolution
uniform sampler2D smp; // prev scene

#define PI 3.141592653589793
#define PI2 PI * 2.

float lPolygonCurve(vec2 p,int n,float f){
 float a = atan(p.x,p.y)+PI;
 float r = PI2/float(n);
 return cos(floor(.5+a/r)*r-a)*(length(p)+f)/(1.+f);
}

void main(void){
	vec2 p = (gl_FragCoord.xy * 2.0 - r) / min(r.x, r.y);
	float l=lPolygonCurve(p,5,.5);
	gl_FragColor = vec4(vec3(l), 1.0);
}
```
## 直線
ディスタンスフィールドに(永久に続く)直線はあるが、基本的に線分(2点で区切られた直線)はない。
ので、結論としては、直線をif文で条件分けして実現するのが一番てっとり早い。
他には、テッセレーションで直線を線分に分割したり、四角形の対向する2辺を極限まで小さくする方法があるが、if文で分岐するのが一番楽でしょう。

線分の仕方もいくつあるが、直線の描画方法にもいくつかの種類がある。
長いので他にまとめました！
* [2次元ディスタンスフィールド線分まとめ](https://qiita.com/7CIT/items/52ebd6421c041ac7024a)

## 楕円

![ellipse.png](https://qiita-image-store.s3.amazonaws.com/0/135381/ac280ca3-3842-bd65-285a-b8528fbc87a4.png)
https://goo.gl/zy5mW6

```glsl
precision mediump float;
uniform vec2  m;       // mouse
uniform float t;       // time
uniform vec2  r;       // resolution
uniform sampler2D smp; // prev scene

float lEllipse(vec2 p, vec2 r, float s) {
    return (length(p / r) - s);
}

void main(void){
	vec2 p = (gl_FragCoord.xy * 2.0 - r) / min(r.x, r.y);
	float l;
	l=lEllipse(p,vec2(.5,.25),1.);
	gl_FragColor = vec4(vec3(l), 1.0);
}
```

## 長方形

![rect.png](https://qiita-image-store.s3.amazonaws.com/0/135381/612cefa8-c690-1327-a7b2-cbf7a4d15248.png)

https://goo.gl/fkTvUm

```glsl
precision mediump float;
uniform vec2  m;       // mouse
uniform float t;       // time
uniform vec2  r;       // resolution
uniform sampler2D smp; // prev scene

float lRect(vec2 p, vec2 size) {  
  vec2 d = abs(p) - size;
  return min(max(d.x, d.y), 0.0) + length(max(d,0.0));
}

void main(void){
	vec2 p = (gl_FragCoord.xy * 2.0 - r) / min(r.x, r.y);
	float l;
	l=lRect(p,vec2(.5,.25));
	gl_FragColor = vec4(vec3(l), 1.0);
}
```

## 角丸長方形

![rrect.png](https://qiita-image-store.s3.amazonaws.com/0/135381/ea80d18c-09d5-0b23-cc66-5cec58a0b6d5.png)

https://goo.gl/Kgjpo5

```glsl
precision mediump float;
uniform vec2  m;       // mouse
uniform float t;       // time
uniform vec2  r;       // resolution
uniform sampler2D smp; // prev scene

float lRoundRect(vec2 p, vec2 size, float radius) {
  vec2 d = abs(p) - size + radius;
  return min(max(d.x, d.y), 0.0) + length(max(d,0.0))- radius;
}

void main(void){
	vec2 p = (gl_FragCoord.xy * 2.0 - r) / min(r.x, r.y);
	float l;
	l=lRoundRect(p,vec2(.5,.25),.1);
	gl_FragColor = vec4(vec3(l), 1.0);
}
```

## 十字星

![ltwincle.png](https://qiita-image-store.s3.amazonaws.com/0/135381/b951144b-6bfe-1494-c422-669976901ce3.png)

https://goo.gl/54RVfF

```glsl
precision mediump float;
uniform vec2  m;       // mouse
uniform float t;       // time
uniform vec2  r;       // resolution
uniform sampler2D smp; // prev scene

float lTwincle(vec2 p){return abs(p.x) * abs(p.y);}

void main(void){
	vec2 p = (gl_FragCoord.xy * 2.0 - r) / min(r.x, r.y);
	float l;
	l=lTwincle(p);
	gl_FragColor = vec4(vec3(l), 1.0);
}
```

## 7セグメント(デジタル)数字

![image.png](https://qiita-image-store.s3.ap-northeast-1.amazonaws.com/0/135381/19d91c1f-c4ef-157f-c0ae-b932de782fec.png)

```glsl
precision mediump float;
uniform vec2  m;       // mouse
uniform float t;       // time
uniform vec2  r;       // resolution
uniform sampler2D smp; // prev scene

float seg(vec2 p,float s,float k){p=abs(p*k);return max(s+p.x,s*.5+p.y+p.x);}
float seg7(vec2 p,int n,float s,float k){
    float l=5.,w=s*.5;
    vec2 q=p.yx,h=vec2(s,0.),v=vec2(-w,w);
    l = (n!=1 && n!=4                ?min(seg(q-h,s,k),l):l);
    l = (n!=1 && n!=2 && n!=3 && n!=7?min(seg(p-v,s,k),l):l);
    l = (n!=5 && n!=6                ?min(seg(p-w,s,k),l):l);
    l = (n!=0 && n!=1 && n!=7        ?min(seg(q  ,s,k),l):l);
    l = (n==0 || n==2 || n==6 || n==8?min(seg(p+w,s,k),l):l);
    l = (n!=2                        ?min(seg(p+v,s,k),l):l);
    l = (n!=1 && n!=4 && n!=7        ?min(seg(q+h,s,k),l):l);
    return l;
}
void main(void){
  vec2 p = (gl_FragCoord.xy * 2.0 - r) / min(r.x, r.y);
  
  float l=seg7(p,int(mod(t,10.)),.8,1.3);
  gl_FragColor = vec4(vec3(l), 1.0);
}
```

* [サンプルはこちら](http://jp.wgld.org/js4kintro/editor/#precision%20mediump%20float%3B%0Auniform%20vec2%20%20m%3B%20%20%20%20%20%20%20%2F%2F%20mouse%0Auniform%20float%20t%3B%20%20%20%20%20%20%20%2F%2F%20time%0Auniform%20vec2%20%20r%3B%20%20%20%20%20%20%20%2F%2F%20resolution%0Auniform%20sampler2D%20smp%3B%20%2F%2F%20prev%20scene%0A%0Afloat%20seg(vec2%20p%2Cfloat%20s%2Cfloat%20k)%7Bp%3Dabs(p*k)%3Breturn%20max(s%2Bp.x%2Cs*.5%2Bp.y%2Bp.x)%3B%7D%0Afloat%20seg7(vec2%20p%2Cint%20n%2Cfloat%20s%2Cfloat%20k)%7B%0A%20%20%20%20float%20l%3D5.%2Cw%3Ds*.5%3B%0A%20%20%20%20vec2%20q%3Dp.yx%2Ch%3Dvec2(s%2C0.)%2Cv%3Dvec2(-w%2Cw)%3B%0A%20%20%20%20l%20%3D%20(n!%3D1%20%26%26%20n!%3D4%20%20%20%20%20%20%20%20%20%20%20%20%20%20%20%20%3Fmin(seg(q-h%2Cs%2Ck)%2Cl)%3Al)%3B%0A%20%20%20%20l%20%3D%20(n!%3D1%20%26%26%20n!%3D2%20%26%26%20n!%3D3%20%26%26%20n!%3D7%3Fmin(seg(p-v%2Cs%2Ck)%2Cl)%3Al)%3B%0A%20%20%20%20l%20%3D%20(n!%3D5%20%26%26%20n!%3D6%20%20%20%20%20%20%20%20%20%20%20%20%20%20%20%20%3Fmin(seg(p-w%2Cs%2Ck)%2Cl)%3Al)%3B%0A%20%20%20%20l%20%3D%20(n!%3D0%20%26%26%20n!%3D1%20%26%26%20n!%3D7%20%20%20%20%20%20%20%20%3Fmin(seg(q%20%20%2Cs%2Ck)%2Cl)%3Al)%3B%0A%20%20%20%20l%20%3D%20(n%3D%3D0%20%7C%7C%20n%3D%3D2%20%7C%7C%20n%3D%3D6%20%7C%7C%20n%3D%3D8%3Fmin(seg(p%2Bw%2Cs%2Ck)%2Cl)%3Al)%3B%0A%20%20%20%20l%20%3D%20(n!%3D2%20%20%20%20%20%20%20%20%20%20%20%20%20%20%20%20%20%20%20%20%20%20%20%20%3Fmin(seg(p%2Bv%2Cs%2Ck)%2Cl)%3Al)%3B%0A%20%20%20%20l%20%3D%20(n!%3D1%20%26%26%20n!%3D4%20%26%26%20n!%3D7%20%20%20%20%20%20%20%20%3Fmin(seg(q%2Bh%2Cs%2Ck)%2Cl)%3Al)%3B%0A%20%20%20%20return%20l%3B%0A%7D%0Avoid%20main(void)%7B%0A%20%20vec2%20p%20%3D%20(gl_FragCoord.xy%20*%202.0%20-%20r)%20%2F%20min(r.x%2C%20r.y)%3B%0A%20%20%0A%20%20float%20l%3Dseg7(p%2Cint(mod(t%2C10.))%2C.8%2C1.3)%3B%0A%20%20gl_FragColor%20%3D%20vec4(vec3(l)%2C%201.0)%3B%0A%7D)

## ハートの距離関数
* [ハートの距離関数](https://qiita.com/7CIT/items/5113587c82fc544c0a58)

## 関連
* [2次元ディスタンスフィールド使い方](https://qiita.com/7CIT/items/ea3b41717323c83ecc35)
* [the book of shader のシェイプサンプル](https://thebookofshaders.com/examples/?chapter=07)
* [クリエイティブコーディングをしよう！【その１】GLSLで遊ぶ](https://trap.jp/post/142/)
* [[連載]やってみれば超簡単！ WebGL と GLSL で始める、はじめてのシェーダコーディング（７）](https://qiita.com/doxas/items/25bb50a3db85129e2980)